
Расчетные таблицы (бюджеты) в ИС-ПРО. Основные приемы работы с РТ
Предварительное замечание:

В папке "Приложения" находятся:

1. База "ООО "Прогресс" - Использование расчетных таблиц".
Назначение - дает возможность реально “пощупать” все описанное в методике. База содержит в себе все рассмотренные в методике файлы отчетов и расчетных таблиц. При этом база НЕ содержит никаких лишних "синих" отчетов, поэтому ее можно распаковывать с включенным параметром "Восстанавливать отчеты пользователя".

2. Файлы отчетов и расчетных таблиц.
Назначение - дает возможность пользователю не распаковывать базу "ООО "Прогресс" - Использование расчетных таблиц", а импортировать рассмотренные в методике отчеты и расчетные таблицы в "свою" базу. Однако следует иметь в виду, что в этих файлах имеются настройки, ориентированные именно на базу "ООО "Прогресс" - Использование расчетных таблиц", а это значит, что без предварительной адаптации под “свою” базу файл может не заработать.
3. Описания других методик по расчетным таблицам.
Здесь лежат doc-файлы с описанием некоторых частных задач, решаемых с помощью РТ.

1. Какие задачи можно решать с помощью РТ
Общие задачи:

Построение полноценной системы бюджетирования:

· поддержка неограниченного количества вариантов бюджета;

· возможность формирования бюджета на основе данных оперативного учета (первичных документов), проводок (бухгалтерский, налоговый или другой план счетов) или вручную;

· возможность детализации структуры бюджета (вплоть до документов);

· настройка распределения косвенных затрат предприятия; 

· возможность контроля исполнения бюджетов;

· возможность выгрузки данных в Excel;

· разграничение прав доступа пользователей.
Подробнее см. файл-приложение “Ведение бюджетов в ИС-ПРО на примере БДР.doc”
Использование РТ в качестве хранилища данных:

Одно из основных направлений использования РТ – это использовать их в качестве хранилища данных или своего рода справочника. При этом в отличие от справочников пользователя в РТ имеются более широкие возможности по заданию ключа (т.е. уникального адреса записи). В справочнике пользователя ключ – это значение длиной до 50-ти знаков, а в расчетной таблице ключ задается тремя параметрами – кодом строки (25 знаков), кодом колонки (25 знаков) и ИД аналитической карточки. При этом аналитическая карточка может иметь до пяти уровней в dos-версии и до десяти в win-версии, и на каждом уровне значение кода аналитики имеет длину 50 знаков.

Практическую пользу от этого можно показать на следующем примере. Предположим, стоит задача хранить (и использовать в отчетности / программе проводок / др.) для номенклатуры 5-ть видов дополнительной описательной информации, каждая из которых может быть длинной до 255-ти знаков. Если делать это в справочниках пользователя, то придется заводить 5-ть справочников пользователя. Если же воспользоваться РТ, то достаточно одной РТ с одной строкой, привязанной к картотеке “Номенклатура”, и пятью столбцами.

Дополнительным преимуществом РТ перед справочниками пользователя в данном случае является интерфейс. В справочнике пользователя придется вводить значение ключа (в данном случае артикула) без видимости того, существует ли номенклатура с таким артикулом и какое наименование она имеет. А в РТ при создании каждой новой записи будет открываться реальный справочник номенклатуры, что делает невозможным ввод несуществующего артикула в качестве ключа. 

Примеры частных задач:

1. Расчет себестоимости в многопередельном производстве по видам продукции.

2. Формирование произвольных баз распределения и калькулирования затрат.

3. Протоколирование работы пользователей в ИС-ПРО (подробнее см. файл-приложение “Протоколирование истории документов РТ.doc”).

4. Обеспечение преемственности данных с течением времени (подробнее см. файл-приложение “Обеспечение преемственности данных с течением времени с помощью РТ.doc”).
5. Автоматическое разнесение сумм в платежах на статьи затрат в зависимости от настройки в РТ соответствий Номенклатура --> статья затрат, Группа номенклатуры --> статья затрат.
6. Специализированные отчеты (баланс, приложения к балансу и др.) на основании проводок (бухгалтерский, налоговый или другой план счетов).
7. Расчет пени по налогам, контроль правильности начисления пени налоговой инспекцией (подробнее см. соответствующую методику).
8. Расчет пени за отгруженную теплоэнергию.
9. Настройка соответствия группа продукции --> журнал документов для контроля правильности выписки документов.

2. Способы заполнения РТ

3. Заполнение РТ изнутри
3.1 Заполнение РТ на основе данных главной книги (проводок)
Рассмотрим для примера следующую задачу:
Имеется 25-ый счет. На нем открыта аналитика по ШПЗ (Подразделение/Статья затрат). Вывести в расчетную таблицу обороты за месяц по дебету 25-ого счета в разрезе Статьей затрат, т.е. без учета Подразделений.

Решение в БД “ООО “Прогресс” – Использование расчетных таблиц”:

В бухгалтерских справках занесены проводки по 25-му счету за июнь 2006 года. В журнале расчетных таблиц “Документы без журнала” настроен шаблон РТ с кодом “01” и наименованием “25-ый счет (факт)”. На основании этого шаблона нужно создать таблицу за июнь 2006 года и, находясь внутри созданной таблицы, нажать Alt-P. В результате будет произведен расчет таблицы.

Ниже приведена ЧАСТЬ_1 программы расчетной таблицы, в которой организуется цикл по аналитическим карточкам указанного счета, по которым есть обороты или сальдо за указанный период. В случае если по текущей карточке имеется оборот по дебету, сумма заносится в расчетную таблицу.
============================== ЧАСТЬ 1 ПРОГРАММЫ РАСЧЕТНОЙ ТАБЛИЦЫ
MySch = "25"

ACC_CALC(MySch,RT_DATPER,RT_DATPER_END)

CdStr1  = "1"

CdStlb1  = "1"

IdKart1 = FTBL_L_ANL(RT_ID,CdStr1) 'ИД картотеки, привязанной к строке CdStr1

'---------- подготовка (подсчет аналитических карточек на счете)

KOL_ZAP = 0

IdAnl=ACC_ANL_FIRST(MySch)

WHILE IdAnl<>0

  KOL_ZAP = KOL_ZAP + 1

  IdAnl=ACC_ANL_NEXT(MySch)

WEND

'---------- запись данных в РТ

INDIC_INIT( "Ожидайте, идет процесс ...",KOL_ZAP)

n = 1

IdAnl=ACC_ANL_FIRST(MySch)

WHILE IdAnl<>0

  ObrDT = ACC_TURN_DT( MySch,IdAnl,"")

  IF ObrDT<>0 THEN

     KD1 = TRIM$(ANLCRD_GET_CD(IdAnl,2))

     IdAnlNew = ANLCRD_C_CRT(IdKart1,KD1,"","","","") 'получаем ИД карточки для одноуровневой картотеки "Статьи затрат", т.к. на 25-м счете открыта другая аналитика (двухуровневая по ШПЗ)

     SumOld = FTBL_GET_SM(RT_ID,CdStr1,CdStlb1,IdAnlNew) 'получаем сумму по данной статье затрат, ранее записанную в РТ по другим подразделениям

     '--- ЗАПИСЬ СУММЫ (ObrDT+SumOld) В ДАННУЮ РАСЧЕТНУЮ ТАБЛИЦУ ПО АДРЕСУ: СТРОКА=CdStr1,КОЛОНКА=CdStlb1,КАРТОЧКА=IdAnlNew

     FTBL_SET_SM(CdStr1,CdStlb1,IdAnlNew,ObrDT+SumOld)

  ENDIF

  IdAnl=ACC_ANL_NEXT(MySch)

  n = n+1

  INDIC_REDRAW(n)

WEND
INDIC_REDRAW(KOL_ZAP) 'прорисовываем 100%

INDIC_TERM()

============================== КОНЕЦ ЧАСТИ 1 ПРОГРАММЫ РАСЧЕТНОЙ ТАБЛИЦЫ

Используемые в данной программе функции из группы “Функции расчетных таблиц”:

1. Запись значения в таблицу производится с помощью функции FTBL_SET_SM:
FTBL_SET_SM(CdStr1,CdStlb1,IdAnlNew,ObrDT + SumOld)

Данная функция заносит значение (ObrDT+ SumOld) в ячейку по адресу:
СТРОКА = CdStr1,  КОЛОНКА = CdStlb1,  КАРТОЧКА = IdAnlNew.

Обратите внимание, что эта функция может применяться только во внутренней программе самой расчетной таблицы. Для программного заполнения РТ из внешней подсистемы (отчета, программы проводки и т.п.) используется функция FTBL_SET_SMX (рассмотрена ниже).

2. Чтение значения из таблицы производится с помощью функции FTBL_GET_SM:
SumOld = FTBL_GET_SM(RT_ID,CdStr1,CdStlb1,IdAnlNew)

С помощью данной функции получаем значение ячейки по адресу:

ИД таблицы = RT_ID,  СТРОКА = CdStr1,  КОЛОНКА = CdStlb1,  КАРТОЧКА = IdAnlNew.
Обратите внимание, что в этой функции в качестве параметра фигурирует “ИД таблицы”. Это значит, что для обращения к данным РТ из внешней подсистемы (отчета, программы проводки и т.п.) используется эта же функция. Однако в случае ее применения во внутренней программе значение “ИД таблицы” указывается с помощью системного поля RT_ID, а в случае ее применения из внешней подсистемы значение “ИД таблицы” нужно получать с помощью специальных функций, которые будут рассмотрены ниже.

3. Получение “ИД аналитической картотеки, привязанной к строке таблицы” производится с помощью функции FTBL_L_ANL:
IdKart1 = FTBL_L_ANL(RT_ID,CdStr1)

Данная функция возвращает “ИД аналитической картотеки”, привязанной к строке CdStr1 таблицы RT_ID. 
По аналогии с предыдущей функцией, в этой функции в качестве параметра фигурирует “ИД таблицы”, и поэтому она может использоваться при обращении к РТ из внешней подсистемы.

Применяется в основном в тех случаях, когда необходимо получить ИД карточки с помощью функции ANLCRD_C_CRT, в параметрах которой фигурирует “ИД аналитической картотеки”.
3.2 Заполнение РТ вручную
Рассмотрим для примера следующую задачу:
Вручную занести в расчетную таблицу плановые значения затрат в разрезе Статьей затрат.
Решение в БД “ООО “Прогресс” – Использование расчетных таблиц”:

В журнале расчетных таблиц “Документы без журнала” настроена РТ с кодом “02” и наименованием “25-ый счет (план)”. Для того чтобы в таблицу было возможно вводить значения вручную, в настройках колонки указан тип колонки “Вводимая”:

Обратите внимание на опцию “Признак перерасчета данных, введенных вручную”. В активном состоянии данная настройка разрешает программную корректировку данных, введенных вручную. Т.е. если в ячейку введено значение вручную, а затем произведена попытка присвоить этой ячейке значение с помощью функции FTBL_SET_SM или FTBL_SET_SMX, то значение успешно присвоится. Если же опция в неактивном состоянии, то введенные вручную данные не могут быть программно изменены. В тоже время, пока в ячейку не введено значение вручную, остается возможность задать его программно.
Чтобы ввести значение в таблицу, нужно, находясь в таблице, “провалиться” в нужную строку (в нашем случае она единственная - “Затраты 25-го счета”) и нажать Insert. Далее в открывшемся окне в поле “Статьи затрат” открыть справочник по F3 и выбрать нужную статью. Затем ввести значение ячейки по столбцу “Сумма затрат”:

После нажатия ОК появится соответствующая запись в таблице:

Обратите внимание, невозможно вручную ввести итог по строке/столбцу. О том, как в расчетных таблицах  подбиваются итоги, смотрите п.3 “Формирование итогов”.  
4. Формирование итогов
Для пояснения того, что в данном случае подразумевается под формированием итогов, приведен следующий пример таблицы:

	
	Колонка 1
	Колонка 2

	Строка 1
	Итог по строке 1, колонке 1
	Итог по строке 1, колонке 2

	Аналит. карточка 1
	Значение 1
	Значение

	Аналит. карточка 2
	Значение 2
	Значение

	…
	…
	Значение

	Аналит. карточка N
	Значение N
	Значение

	Строка 2
	Итог по строке 2, колонке 1
	Итог по строке 2, колонке 2

	Аналит. карточка A
	Значение
	Значение

	Аналит. карточка B
	Значение
	Значение

	…
	Значение
	Значение

	Аналит. карточка X
	Значение
	Значение


Формирование Итогов не происходит само собой, т.е. значения полей “Итог по строке 1, колонке 1” (равен Значение 1 + Значение 2 + … + Значение N), “Итог по строке 1, колонке 2” и тому подобных нужно специальным образом рассчитывать и заносить. Как это делается? 
Пример формирования итогов в БД “ООО “Прогресс” – Использование расчетных таблиц”:

ВАРИАНТ 1

Воспользуемся уже рассмотренной таблицей, а именно РТ с кодом “01” и наименованием “25-ый счет (факт)”. Для формирования итогов здесь используется ЧАСТЬ_2 программы. 
В этой части программы организуется цикл по:

- всем строкам таблицы
  - всем столбцам таблицы
    - всем аналитическим карточкам строки

В момент, когда достигнута последняя аналитическая карточка по данной строке, в ячейку “Итог по строке i, колонке j” заносится накопленное значение.  
============================== ЧАСТЬ 2 ПРОГРАММЫ РАСЧЕТНОЙ ТАБЛИЦЫ
'--- универсальная процедура подсчета итогов по всем строкам/столбцам

STR = FTBL_L_FIRST(RT_ID)

WHILE STR<>""

  STLB = FTBL_C_FIRST(RT_ID)

  WHILE STLB<>""

    SumItog = 0

    IdAnl = FTBL_A_FIRST(RT_ID,STR)

    WHILE IdAnl<>0

      SumTek = FTBL_GET_SM(RT_ID,STR,STLB,IdAnl)

      SumItog = SumItog + SumTek

      IdAnl = FTBL_A_NEXT(RT_ID,STR)

    WEND

    FTBL_SET_SM(STR,STLB,0,SumItog)

    STLB = FTBL_C_NEXT(RT_ID)

  WEND

  STR = FTBL_L_NEXT(RT_ID)

WEND

============================== КОНЕЦ ЧАСТИ 2 ПРОГРАММЫ РАСЧЕТНОЙ ТАБЛИЦЫ
Используемые в данной программе функции из группы “Функции расчетных таблиц”:

Цикл по всем строкам таблицы организуется с помощью функций:

STR = FTBL_L_FIRST(RT_ID)

STR = FTBL_L_NEXT(RT_ID)
Цикл по всем столбцам таблицы организуется с помощью функций:

STLB = FTBL_C_FIRST(RT_ID)

STLB = FTBL_C_NEXT(RT_ID)
Цикл по всем аналитическим карточкам строки организуется с помощью функций:

IdAnl = FTBL_A_FIRST(RT_ID,STR)

IdAnl = FTBL_A_NEXT(RT_ID,STR)

Чтение значения текущей ячейки производится с помощью функции FTBL_GET_SM (рассмотрена выше):
SumTek = FTBL_GET_SM(RT_ID,STR,STLB,IdAnl)

Запись итога в таблицу в приведенной программе производится с помощью функции FTBL_SET_SM (рассмотрена выше):
FTBL_SET_SM(STR,STLB,0,SumItog)
Для записи итога из внешней подсистемы (отчета, программы проводки и т.п.) нужно воспользоваться функцией FTBL_SET_SMX (рассмотрена ниже).
Обратите внимание, значение параметра “ИД аналитики” равно нулю. Именно нулевое значение этого параметра обуславливает записать значения в итог по данной строке/столбцу. 

Обратите внимание! Приведенная процедура формирования итогов является полностью универсальной и может использоваться в любой расчетной таблице.
ВАРИАНТ 2
Воспользуемся уже рассмотренной таблицей, а именно РТ с кодом “02” и наименованием “25-ый счет (план)”. Здесь для формирования итога используется программа, формирующая итог не по всем строкам/столбцам, а только по одной строке/столбцу. Такой подход может быть целесообразным, когда подсчет итогов по всем строкам/столбцам по какой-либо причине не нужен или нужен, но только в другой момент.
============================== ПРОГРАММА РАСЧЕТНОЙ ТАБЛИЦЫ
'--- Блок подбивает итог по указанным строке/столбцу

STR  = "1"     'указывается код строки

STLB = "1"    'указывается код столбца

ZNACH = 0   'инициализация переменной

ID_ANL = FTBL_A_FIRST(RT_ID,STR)

WHILE ID_ANL <> 0

      ZNACH = ZNACH + FTBL_GET_SM(RT_ID,STR,STLB,ID_ANL)

      ID_ANL = FTBL_A_NEXT(RT_ID,STR)

WEND

FTBL_SET_SM(STR,STLB,0,ZNACH)
============================== КОНЕЦ ПРОГРАММЫ РАСЧЕТНОЙ ТАБЛИЦЫ
Используемые в этой программе функции из группы “Функции расчетных таблиц” уже рассмотрены выше. 

Обратите внимание! Приведенная процедура формирования итогов является почти универсальной, для ее применения в другой таблице достаточно указать “свои” значения кодов строки и столбца.

5. Программное удаление записей таблицы (очищение)
Удаление конкретной записи по аналитике производится с помощью функции FTBL_A_DEL_ROW (ИД таблицы, Код строки, ИД аналитической карточки). 
Рассмотрим пример в БД “ООО “Прогресс” – Использование расчетных таблиц”:

Воспользуемся уже рассмотренной таблицей, а именно РТ с кодом “01” и наименованием “25-ый счет (факт)”. Ниже приведена программа удаления всех аналитических карточек во всех строках таблицы:

'--- очищение (удаление всех аналитических карточек во всех строках таблицы)

STR = FTBL_L_FIRST(RT_ID)

WHILE STR<>""

   IdAnl = FTBL_A_FIRST(RT_ID,STR)

   WHILE IdAnl<>0

      FTBL_A_DEL_ROW(RT_ID,STR,IdAnl)

      IdAnl = FTBL_A_FIRST(RT_ID,STR)  ‘еще раз идем на первую запись, т.к. после удаления
                                                                   ‘функция FTBL_A_NEXT() возвращает ноль
   WEND

   STR = FTBL_L_NEXT(RT_ID)

WEND
Обратите внимание! Приведенная процедура удаления является полностью универсальной и может использоваться в любой расчетной таблице.

6. Обращение к РТ извне (из отчета / программы проводки / другой расчетной таблицы …)
6.1 Обращение к РТ извне с целью зачитать данные из таблицы
Задача обращения к таблице извне сводится к двум основным подзадачам:

1. Получение значения ИД таблицы, к которой нужно обратиться (для указания в функциях в качестве параметра)

2. Непосредственные манипуляции с этой таблицей (чтение, запись и т.п.)
Для получения значения ИД таблицы используются функции из группы “Функции навигации по реестру таблиц”: 

Можно выделить два основных варианта получения ИД таблицы:

1. Использовать функцию FTBL_FIND( ШаблонТаблицы,ДатаПериода,Номер,КодПодр,Квартал,Год);
2. Производить сканирование имеющихся таблиц с помощью функций навигации (FTBL_FIRST, FTBL_FIRSN …) до тех пор, пока путем такого перебора не найдется таблица с нужными значениями параметров (даты, номера …). При этом дата, номер и другие параметры текущей таблицы определяются с помощью функций FTBL_GET_DATE (ID_таблицы), FTBL_GET_NMR (ID_таблицы) и т.п.

Функции навигации достаточно просты в применении, поэтому не будем их подробно рассматривать.

А что касается функции FTBL_FIND, то здесь имеются нюансы.

Рассмотрим для примера следующую задачу и ее решение в БД “ООО “Прогресс” – Использование расчетных таблиц”:
В журнале расчетных таблиц “Документы без журнала” настроены таблицы:

- с кодом “01” и наименованием “25-ый счет (факт)”, данные в ней формируются на основе проводок;
- с кодом “02” и наименованием “25-ый счет (план)”, данные в нее вносятся вручную;
- с кодом “03” и наименованием “25-ый счет (отклонения)”.
Необходимо заполнить таблицу с кодом “03” следующим образом:
- значение колонки “Факт” заполнить на основе данных таблицы “01”.

- значение колонки “План” заполнить на основе данных таблицы “02”
- обращаться к таким таблицам “01” и “02”, которые имеют такие же номер и дату, что и таблица “03”.
- рассчитать значение колонки “Отклонение” путем вычитания факта от плана;

Для решения этой задачи в таблице “25-ый счет (отклонения)” написана программа.

Приведем текст программы по каждой подзадаче.

Программа заполнения значения колонки “Факт” на основе данных таблицы “01”:

Шаг 1 - Получение значения ИД таблицы “25-ый счет (факт)”
FTBL_SEL_JRN( "0" )
ID_TBL = FTBL_FIND("01",RT_DATPER,RT_NMR,"",0,0)

Функция FTBL_SEL_JRN устанавливает текущий журнал шаблонов расчетных таблиц. В данном случае значение параметра “Код журнала” равно нулю, так как нужная таблица находится в журнале “Документы без журнала”. Перед вызовом функции FTBL_FIND рекомендуется всегда устанавливать журнал.

Пояснения к значениям параметров функции FTBL_FIND:

ШаблонТаблицы = "01" – без комментариев (это код шаблона таблицы “25-ый счет (факт)”).

Поскольку, согласно условию задачи, дата и номер данной таблицы равны дате и номеру сканируемой таблицы, то в значения параметров функции подставляем значения даты и номера данной таблицы:
ДатаПериода = RT_DATPER 'дата начала периода таблицы

Номер       = RT_NMR             'номер таблицы

Поскольку в таблице “25-ый счет (факт)” отсутствуют такие параметры как Код подразделения, Квартал и Год, то в параметрах функции FTBL_FIND они указываются так:

КодПодр     = ""

Квартал     = 0

Год         = 0

Замечание 1:

В случаях, когда нельзя программно задать дату, номер или какой-то другой параметр таблицы, эти значения можно запрашивать у пользователя. Например, такая схема актуальна, когда имеется таблица, в которую записываются данные из отчета. При запуске отчета возникает запрос даты и номера таблицы, в которую нужно записать данные.

Замечание 2:

Если таблица имеет параметр дату и необходимо с помощью функции FTBL_FIND определить ее ИД, то значение параметра даты обязательно должно быть равно первому дню месяца, к которому относится данная таблица. Поясним на примере. Есть таблица за период Июнь, 2006. В программе калькуляции документа расходной накладной мы обращаемся к таблице. Если в параметрах FTBL_FIND мы укажем в качестве параметра даты дату документа, то для документа с датой 05.06.2006 функция FTBL_FIND вернет ноль. Поэтому в подобных случаях в качестве параметра даты нужно ставить не саму дату, а функцию MONTH_BEG( Дата ), преобразующую дату в дату первого дня месяца.

Если таблица НЕ имеет параметр дату, то в значении этого параметра указывается ноль.

Шаг 2 – Заполнение значения колонки “Факт” на основе данных таблицы “01”

STR       = "1"

STBL_SCAN = "1"

STBL_REC  = "1"

IdAnl = FTBL_A_FIRST(ID_TBL,STR)

WHILE IdAnl<>0

   SumTek = FTBL_GET_SM(ID_TBL,STR,STBL_SCAN,IdAnl)

   FTBL_SET_SM(STR,STBL_REC,IdAnl,SumTek)

   IdAnl = FTBL_A_NEXT(ID_TBL,STR)

WEND
В этой части программы организуется цикл по аналитическим карточкам строки “1” таблицы “01”. По каждой карточке зачитывается значение (с помощью функции FTBL_GET_SM) и записывается (с помощью FTBL_SET_SM) в колонку “Факт” таблицы “03”.
Программа заполнения значения колонки “План” на основе данных таблицы “02”:
(оба шага выполняются аналогично предыдущей подзадаче)

Шаг 1 - Получение значения ИД таблицы “25-ый счет (план)”

FTBL_SEL_JRN( "0" )

ID_TBL = FTBL_FIND("02",RT_DATPER,RT_NMR,"",0,0)

Шаг 2 – Заполнение значения колонки “План“ на основе данных таблицы “02”

STR       = "1"

STBL_SCAN = "1"

STBL_REC  = "2"

IdAnl = FTBL_A_FIRST(ID_TBL,STR)

WHILE IdAnl<>0

   SumTek = FTBL_GET_SM(ID_TBL,STR,STBL_SCAN,IdAnl)

   FTBL_SET_SM(STR,STBL_REC,IdAnl,SumTek)

   IdAnl = FTBL_A_NEXT(ID_TBL,STR)

WEND
Программа расчета колонки “Отклонение”:

STR = "1"

IdAnl = FTBL_A_FIRST(RT_ID,STR)

WHILE IdAnl<>0

   SumPlan = FTBL_GET_SM(RT_ID,STR,"1",IdAnl)

   SumFakt = FTBL_GET_SM(RT_ID,STR,"2",IdAnl)

   FTBL_SET_SM(STR,"3",IdAnl,SumPlan-SumFakt)

   IdAnl = FTBL_A_NEXT(RT_ID,STR)

WEND

В этой части программы организуется цикл по аналитическим карточкам строки “1” таблицы “03”, зачитываются значения плана и факта, разница между ними заносится в колонку “3”.

6.2 Обращение к РТ извне с целью записать данные в таблицу
Как уже было сказано выше, при обращении к РТ извне для чтения или записи, вначале нужно получить ИД таблицы, а затем производить те или иные действия с помощью специальных функций.
Порядок получения ИД таблицы в общем виде рассмотрен выше.
Для записи данных в таблицу извне используется функция FTBL_SET_SMX (ID_таблицы, "КодСтр", "КодКол", ID_аналит, Значение). По сути, она аналогична функции FTBL_SET_SM (рассмотрена выше), за исключением того, что в качестве дополнительного параметра фигурирует ИД таблицы.
Рассмотрим для примера следующую задачу:
Получить информацию о расходе сырья в производство за период в разрезе Подразделение/Сырье/Продукт в виде План/Факт/Отклонение. 
Решение в БД “ООО “Прогресс” – Использование расчетных таблиц”:

В журнале расчетных таблиц “Произ-во” настроена РТ с кодом “РасхСрПр” и наименованием “Расход сырья ЗА ПЕРИОД План-Факт-Откл”. В подсистеме “Управление производством” / далее пункт меню “Отчеты” на основе шаблона 1194 настроен отчет “WIN Запись в РТ Расхода сырья ЗА ПЕРИОД”, который записывает в РТ нужную информацию. 
В данном примере процедура записи данных из отчета в РТ имеет следующую последовательность:
1. Создаем таблицу, указываем номер.

2. Запускаем отчет за период с 01.09.2000 по 30.09.2000.

3. При запуске отчета возникнет запрос номера таблицы, в которую нужно записать данные. Значение этого номера необходимо для получения ИД таблицы с помощью функции FTBL_FIND.

В результате появится отчет с сообщением:

А в расчетной таблице появятся записанные данные:
[image: image11.png]Omuém Cepsuc

Tasnuya: | 01

HaumeHogaHue | ¢ | Hopma Pakm OmknoHeHUe

Pacxog ctipes,

OminoHeHUA 900.00 620.00 280.00
P Dbs o 8 8 8

B 0.00

=

F1:Momome F10:MeHm Esc:Bbxog [ RU [23,/06.,/06 [15:10





Ниже приведены программы отчета “WIN Запись в РТ Расхода сырья ЗА ПЕРИОД”. Большинство функций, которые в них используются, были рассмотрены выше. Если какая-то из функций Вам непонятна или неизвестна, то откройте приложение №4 (pril4ru.doc или pril4ua.doс) документации к системе и выполните поиск нужной функции.
Программа инициализации отчета:

 PICTURE_ADD("Введите номер расчетной таблицы:","S",8)

 PICTURE_READ()

 R_NMR = NUMTOSTR(PICTURE_GET())

 rem Находим таблицу

 FTBL_SEL_JRN("Произ-во")

 IDT_ = FTBL_FIND("РасхСрПр",0,R_NMR,"",0,0) 'Расход Сырья за Период

 IF IDT_ <> 0 THEN

      R_IDT=IDT_

      R_IDA=FTBL_L_ANL(R_IDT,"001") 'зачитываем ИД аналитической картотеки, привязанной к строке таблицы

      '--- очищение (удаление всех аналитических карточек во всех строках таблицы)

      STR = FTBL_L_FIRST(R_IDT)

      WHILE STR<>""

       IdAnl = FTBL_A_FIRST(R_IDT,STR)

        WHILE IdAnl<>0

           FTBL_A_DEL_ROW(R_IDT,STR,IdAnl)

           IdAnl = FTBL_A_FIRST(R_IDT,STR)

        WEND

        STR = FTBL_L_NEXT(R_IDT)

      WEND

 ENDIF

 IF R_IDT<=0 THEN  'если таблица не найдена, прерываем отчет

    MESSAGE("Р/таблицы "+RTRIM$(R_NMR)+" не найдено!")

    REP_BREAK()

 ENDIF
Программа, привязанная к детальной строке отчета:

TYPE_ANLSPR = 2  ' Аналитический справочник

 CD_ANLSPR_1 = 33 ' Аналитический справочник: Номенклатура (аналит.справ-к)

 KD_1 = NUMTOSTR(rsNomArt)           '!

       'добавление новых записей в аналит. справ.

       IF NUMTOSTR (GET_SPR_VALUE( TYPE_ANLSPR, CD_ANLSPR_1, KD_1)) = "" THEN

          PUT_SPR_VALUE( TYPE_ANLSPR, CD_ANLSPR_1,KD_1,rsNomNmAlt)

       ENDIF

 IDK=ANLCRD_C_CRT(R_IDA,PRDRPT_PDR_CD,PRDRP_GP_ART,KD_1,"","")

 IF IDK>0 THEN

   'записываем в строку Расход сырья, Отклонения

   IF (PRDRP_MAT_NRMP <> 0 OR PRDRP_MAT_FCTP <> 0) AND PRDRP_MAT_NRMP<>PRDRP_MAT_FCTP THEN

      S001_OLD = 0 'для накоплений количества расхода

      S001_OLD = FTBL_GET_SM(R_IDT,"001","001",IDK)

      FTBL_SET_SMX(R_IDT,"001","001",IDK,PRDRP_MAT_NRMP+S001_OLD)

      S002_OLD = 0

      S002_OLD = FTBL_GET_SM(R_IDT,"001","002",IDK)

      FTBL_SET_SMX(R_IDT,"001","002",IDK,PRDRP_MAT_FCTP+S002_OLD)

      S003_OLD = 0

      S003_OLD = FTBL_GET_SM(R_IDT,"001","003",IDK)

      FTBL_SET_SMX(R_IDT,"001","003",IDK,(PRDRP_MAT_NRMP-PRDRP_MAT_FCTP)+S003_OLD)

   ENDIF

   'записываем в строку Расход сырья, Всего

   IF PRDRP_MAT_NRMP <> 0 OR PRDRP_MAT_FCTP <> 0 THEN

      S001_OLD = 0 'для накоплений количества расхода

      S001_OLD = FTBL_GET_SM(R_IDT,"002","001",IDK)

      FTBL_SET_SMX(R_IDT,"002","001",IDK,PRDRP_MAT_NRMP+S001_OLD)

      S002_OLD = 0

      S002_OLD = FTBL_GET_SM(R_IDT,"002","002",IDK)

      FTBL_SET_SMX(R_IDT,"002","002",IDK,PRDRP_MAT_FCTP+S002_OLD)

      S003_OLD = 0

      S003_OLD = FTBL_GET_SM(R_IDT,"002","003",IDK)

      FTBL_SET_SMX(R_IDT,"002","003",IDK,(PRDRP_MAT_NRMP-PRDRP_MAT_FCTP)+S003_OLD)

   ENDIF

 ENDIF

 IF IDK=0 THEN

    Srt1 = "Не создана аналит. карточка.\nВероятно, для ном-ры не определен аналит. код.\n"

    MESSAGE(Srt1+NUMTOSTR(R_IDA)+" - "+PRDRPT_PDR_CD+", "+KD_1+", "+PRDRP_GP_ART+" !!!")

 ENDIF
Программа завершения отчета:

FTBL_PREPARE()

FTBL_RUN( R_IDT )

'отражаем, за какой период был рассчитан отчет

 FTBL_SET_SMX(R_IDT,"001","004",0,"c "+FDATE_STR(PRDRPS_DAT_BEG,6)+ " по "+FDATE_STR(PRDRPS_DAT_END,6))

7. Вывод РТ на печать
Имеется два основных варианта вывода РТ на печать:
1. Находясь внутри конкретной таблицы нажать F9 и запустить отчет “1152 Расчетная таблица”. 
Этот вариант в большинстве случае мало “жизнеспособен”. Во-первых, внешний вид таблицы, сформированный этим отчетом, зачастую неудовлетворителен, а, во-вторых, если таблица содержит много строк, столбцов и аналитик, то этот отчет вообще не сформируется по причине большого объема данных (выведется сообщение об ошибке и отчет прервется).
2. Воспользоваться отчетом 816, который отличается следующими особенностями:

a. Вызывается изнутри таблицы, из реестра таблиц или вообще из другой подсистемы (например, “Главная книга”); 
b. Выводит данные именно по тем строкам и столбцам таблицы и с такой группировкой, которые нужны в данном конкретном случае. Другими словами, в настройках 816-го отчета возможностей “маневрирования” значительно больше.

c. 816-ый отчет напрямую связан с расчетными таблицами, т.е. если зайти в конкретную таблицу и запустить 816-ый отчет, то программа отчета будет знать, из какой таблицы запустили отчет и корректно возвращать значения полей-реквизитов РТ (ИД таблицы, номер, период и т.п.).
Рассмотрим для примера следующую задачу и ее решение в БД “ООО “Прогресс” – Использование расчетных таблиц”:
В журнале расчетных таблиц “Документы без журнала” настроена РТ с кодом “Ном-ра” и наименованием “Номенклатура”. Нужно вывести содержимое этой таблицы на печать.

Решение. На основе 816-го шаблона настроен отчет “WIN Вывод данных РТ на печать (пример)”: 

В результате запуска сформируется отчет вида:

Ниже приведены программы отчета.

Программа инициализации отчета:

REP_SETSORT(1)

 R_IDT = RT_ID  'при запуске 816-го отчета изнутри таблицы заполняются поля-реквизиты РТ

 IF R_IDT <= 0 THEN

    '--- вариант получения ИД таблицы при запуске отчета из другого модуля (извне таблицы)

     CD_SHABL = "Ном-ра"

     FTBL_SEL_JRN( "0" )

     R_IDT = FTBL_FIND(CD_SHABL,0,"","",0,0)

     IF R_IDT <= 0 THEN

        MESSAGE( "Таблица не найдена! Отчет прерван." )

        REP_BREAK()

     ENDIF

 ENDIF

Программа генерации отчета:

'подсчет кол-ва записей в строке 01

 ID_ANL = FTBL_A_FIRST(R_IDT,"01")

 i = 1

 WHILE ID_ANL <> 0

       ID_ANL = FTBL_A_NEXT(R_IDT,"01")

       i = i + 1

 WEND

 INDIC_INIT( "формирование отчета...", i )

 n = 0

 ID_ANL = FTBL_A_FIRST(R_IDT,"01")

 WHILE ID_ANL <> 0

  KD_NOM = ANLCRD_GET_CD(ID_ANL,1)

  NM_NOM = ANLCRD_GET_NM(ID_ANL,1)

  CMT    = FTBL_GET_SM(R_IDT,"01","01",ID_ANL)

  '--- формируем детальную строку

  DETAIL(1)

  '--- переходим к следующей записи

 ID_ANL = FTBL_A_NEXT(R_IDT,"01")

 n = n + 1

 INDIC_REDRAW(n)

 WEND

 INDIC_TERM()
Обратите внимание на то, как формируется детальная строка – внутри цикла по строке таблицы для каждой очередной аналитической карточки вызывается команда генерации детальной строки DETAIL(1).
8. Редактирование шаблона РТ, на основании которого уже созданы таблицы
Как только на основании шаблона РТ создана хотя бы одна таблица, этот шаблон становится недоступным для ручного редактирования. Т.е. невозможно изменение каких-либо существенных настроек таблицы, кроме программы и наименований строк и столбцов. Поэтому в процессе настройки шаблона нужно сразу ориентироваться на то, что произведенные настройки будут стабильными.
Вместе с тем, всегда имеется возможность программного редактирования шаблона таблицы – создания и удаления строк и столбцов (обратите внимание, нет возможности изменения настройки строк и столбцов, а только создания и удаления). Для этого используется группа функций “Функции для колонок и строк фин.таблиц”:

Использование данных функций имеет следующие особенности:

1. Добавление или удаление распространяется на все ранее созданные на основе этого шаблона таблицы. Т.е. если на основании данного шаблона имеется несколько заполненных таблиц и затем с помощью этих функций удаляется одна из колонок, то во всех этих таблицах эта колонка тоже удалиться, данные будут безвозвратно потеряны.

2. Перед добавлением или удалением вызывается функция “Блокировать шаблон”, а после – “Разблокировать шаблон”.  
Пример добавления колонок: 

Пусть имеется шаблон таблицы с кодом “1” в журнале “Документы без журнала”. Нижеприведенная программа привязана к детальной строке отчета, вызываемого по F9 в аналитических справочниках (отчет выводит все статьи аналитического справочника). В результате отработки отчета на каждую статью справочника в РТ возникнет колонка с соответствующим кодом и наименованием.
FTBL_EDT_BEG(1,0)

FTBL_ADD_COL(TRIM$(SPR_CD),SPR_NM,1,2,0,0,0)

FTBL_EDT_END(0)
9. Экспорт/Импорт шаблонов РТ

Шаблоны таблиц можно экспортировать и импортировать с помощью команд Alt-E и Alt-I соответственно. В случае импорта имеется один нюанс. Прежде чем использовать импортированный шаблон, необходимо проверить настройку привязанных аналитических картотек к строкам таблицы. Поскольку в разных базах имеются совершенно разные аналитические картотеки, то зачастую при проверке привязки аналитической картотеки к строке там обнаруживается совершенно другая картотека, чем та, которая привязана в базе, откуда делался экспорт шаблона.

10. Дополнительные замечания

10.1 Взаимосвязи между таблицами

При построении системы бюджетирования актуален вопрос о том, как настраиваются взаимосвязи между таблицами (при расчете таблицы вначале рассчитываются другие таблицы, от которых она зависит). Для получения сведений по этому вопросу и многим другим рекомендуем ознакомиться c:
· подразделом 14.1 “Модуль Бюджеты” документации к системе  (в электронном виде документация лежит в папке установки ИС-ПРО / далее папка DOC).
· полным списком функций из группы “Функции расчетных таблиц” (приложение №4 документации к системе). Даже беглое ознакомление с описанием этих функций существенно расширит Ваши представления об имеющемся инструментарии в работе с таблицами. 
10.2 Настройка внешнего вида

Относительно возможностей настройки внешнего вида таблицы стоит отметить следующие моменты:

· для быстрого ручного редактирования (не создания, а только редактирования) таблицы используется команда “Корректировка таблицы” по Ctrl-F4. Режим доступен только внутри строки таблицы, т.е. вначале нужно “провалиться” в строку, а затем включать этот режим.

· для изменения представления аналитики используется команда “Вид/Режим отображения аналитики”/далее выбирается один из вариантов “Код и наименование/Код/Наименование”. Этот режим особенно полезен, когда длина аналитического кода достаточно велика и его значение не информативно для пользователя.















Окно ввода новой позиции





Уже введенные позиции











Здесь две строки. 


В строке “Расход сырья, Отклонения” выведены только те расходы сырья, факт которых отличается от плана.


В строке “Расход сырья, Всего” выведен весь расход сырья.


(подробнее смотрите в базе-приложении ”Прогресс”)











В программе расчетной таблицы в качестве системных переменных имеется ряд полей-реквизитов РТ, в том числе “ИД таблицы”.


Эти поля возвращают значение только при обращении к ним из внутренней программы.





Указан тип колонки “Вводимая”.


“Признак перерасчета данных, введенных вручную” НЕ активен. 





ИзНУТРИ:


- программно (инициируется запуск программы самой расчетной таблицы по команде Alt-P)


- вручную (значение ячеек вводится с клавиатуры)





Извне:


- программно из внешней подсистемы (отчета, программы проводки и т.п.);


- программно из другой расчетной таблицы.

















Расчетная таблица

